


LED センサー基板を動かす


●LED センサー基板

IchigoJam のピンソケットに差しこんで使う基板です。

LED×8 個、光センサー(フォトトランジスタ)、ブザーがのっています。



回路図は以下のとおりです。




●LED を光らせる

まずは LED を光らせてみましょう。
 ダイレクトモードで、以下のプログラムを打ちます。

```
OUT 1, 1
OUT 2, 1
OUT 3, 1
OUT 4, 1
OUT 5, 1
OUT 6, 1
OUT 7, 1
OUT 8, 1
```

8 個の LED が光ります。



```
OUT 1, 0
OUT 2, 0
OUT 3, 0
OUT 4, 0
OUT 5, 0
OUT 6, 0
OUT 7, 0
OUT 8, 0
```

LED が消えます。

OUT(アウト)命令は、出力ポートに値を出力する命令です。

```
OUT 1, 1
    ポート  値
    番号
```

ポート番号	出力するポートの番号(1~6)。
値	デジタル出力なので、「0」か「1」を指定する。

LED を光らせるには「OUT (ポート番号), 1」と入力して、OUT ポートに「1」(電圧が ON)を出力します。LED は抵抗を通して GND(電源のマイナス)につながっているので、電流が流れて LED が光ります。

「OUT (ポート番号), 0」と入力して、OUT ポートに「0」を出力すると、GNDも電圧が「0」(OFF)なので、電流が流れず、LED は消えます。

●LED を連続で光らせる

LED を連続で光らせるプログラムを作ります。
 まず、1 番の LED を光らせるプログラムを入力します。

10 P=1	変数 P を 1 にする
20 OUT P, 1	LED を光らせる
30 WAIT 6	10 分の 1 秒待つ
40 OUT P, 0	LED を消す

RUN 命令でプログラムを実行してみましょう。1 番の LED が 10 分の 1 秒光ります。
 WAIT 命令の数字を変えて、光る時間をいろいろ変えてみましょう。


今回は**変数**(へんすう)Pを使って、出力するポートを指定しています。
 変数は、数字を入れる箱(いれもの)だと思ってください。
 小学生の人は算数で習う「□」(四角)、中学生以上の人は「x」などの文字と思えばいいです。

さて、同じように 2 番～8 番の LED を光らせるのですが、8 個分のプログラムを入力するのは大変です。

変数 P を自動的に 1～8 まで変化させて、**くりかえし**のプログラムにします。

10 FOR P=1 TO 8	変数 P を 1 から 8 まで、1 ずつ増やす
20 OUT P, 1	
30 WAIT 6	
40 OUT P, 0	
50 NEXT	FOR にもどってくりかえし

RUN 命令でプログラムを実行してみましょう。1 番～8 番の LED が順に光ります。



新しく追加した FOR(フォー)、NEXT(ネクスト) 命令について説明します。

```
FOR   P=1   TO   8   STEP   1
      変数の   最後の値   変化させる値
      最初の値
```

変数の最初の値	指定した変数へ最初にセットする値。
最後の値	この値まで変数を変化させる。
変化させる値	この値ずつ変数を変化させる。 STEP 以下を省略すると1 ずつ増やす。

```
NEXT  ここまでくり返し
```

「FOR P=1 TO 8」とすると、変数 P を1から始めて 8 になるまで1ずつ増やして、「NEXT」までのプログラムをくり返し実行します。

最後に GOTO 命令を追加して、最初にもどってくりかえすようにしましょう。

```
10 FOR P=1 TO 8
20 OUT P,1
30 WAIT 6
40 OUT P,0
50 NEXT
60 GOTO 10
```

10 行にもどる



RUN 命令でプログラムを実行してみましょう。LED が 1 番～8 番までくり返し光ります。

★できる人は

FOR 命令の最初の値、最後の値、変化させる値をいろいろ変えて、LED がどう光るか試してみましょう。

●光センサーを使う

光センサーを使って、周りの明るさによって LED を ON/OFF してみましょう。
 今回は光センサーとして、**フォトランジスタ**を使います。
 フォトランジスタは、光が当たると電流が流れる部品です。




まず、光センサーがちゃんと動作するか確認しましょう。
 以下のプログラムを 1 行追加します。

```

10 FOR P=1 TO 8
15 A=ANA(2):?A
20 OUT P,1
30 WAIT 6
40 OUT P,0
50 NEXT
60 GOTO 10
    
```

光センサーの入力を読み取って画面に表示

RUN 命令でプログラムを実行してみましょう。
 画面に数字が連続で表示されます。まわりの明るさによりますが、
 だいたい 900 を越えたくらいの値になるはずです。



光センサーに指をおいて、かげにしてみましょう。
 数字が 100 以下くらいに小さくなります。(まわりの明るさによって数字は変わります)

「ANA()」は、アナログ入力を読み取る関数です。

ANA(2)
 ポート番号

ポート番号は、2(IN2 端子)、5~8(OUT1~4 端子)、0 か 9(BTN 端子)が使える。
 返ってくる値は、0(0V)~1023(3.3V)。


光センサーで明るさを読み取って、「暗くなったら LED が光る」プログラムにしてみましょう。

```

10 FOR P=1 TO 8
15 A=ANA(2):?A
20 IF A<300 THEN OUT P,1
30 WAIT 6
40 OUT P,0
50 NEXT
60 GOTO 10
    
```

光センサーの入力が 300 より
小さかったら、LED を光らせる

RUN 命令でプログラムを実行してみましょう。
フォトトランジスタを指で暗くすると、LED
が光ります。



20 行で「光センサーからの入力値が 300 より小さかったら、LED を光らせる」という条件判断
をしています。

<code>IF A<300</code>	<code>THEN OUT P,1</code>	<code>ELSE</code>	<code>...</code>
条件式	条件が成り立つ 時に実行	条件が成り立たない 時に実行	

条件式	判断したい条件を設定する。
条件が成り立つ 時に実行	条件式の条件が成り立った時に、この命令を実行する。
条件が成り立た ない時に実行	条件式の条件が成り立たない時に、この命令を実行する。 必要ない時は ELSE 以下は省略できる。

「300」の数値をいろいろ変えて試してみましょう。

周囲の明るさによっては、値を調整しないと、LED を ON/OFF できません。

★できる人は

逆に、「周囲が明るかったら LED を光らせる」(暗くなったら消える)ように、プログラムを改造
してみましょう。

★できる人は

LED センサー基板を使ったプログラムを紹介します。入力して動かしてみてください。

●照度計

光センサーを明るくしたり暗くしたりすると、メーターのように LED が光ります。

```
10  ^*Luminometer
20  A=ANA(2)
30  FOR P=1 TO 8
40  OUT P,A>P*100
50  NEXT
60  GOTO 20
```

「^」は、Enter キーのとなりにあります

●ルーレット

実行して、IchigoJam のボタンを押すと、8 個の LED がランダムに光って止まります。
ボタンを押すとリプレイします。

```
10  ^*Roulette
20  IF !BTN( ) CONT
30  FOR I=1 TO 10
40  GSB 120
50  WAIT 6
60  NEXT
70  FOR I=1 TO 5
80  GSB 120
90  WAIT 20
100 NEXT
110 GOTO 20
120 OUT 0
130 BEEP 10,2
140 P=RND(8)+1
150 OUT P,1
160 RTN
```

●バーサライタ (POV)

実行すると 8 個の LED が細かく光ります。

IchigoJam を手に持って左右に振ると、眼の残像効果でハート形が見えます。

```

10  `*POV
20  CLV:OUT 8,0
30  [0]=`00001110
40  [1]=`00011111
50  [2]=`00111111
60  [3]=`01111110
70  [4]=`11111100
80  [5]=`01111110
90  [6]=`00111111
100 [7]=`00011111
110 [8]=`00001110
120 L=9
130 FOR I=0 TO L
140 OUT [I]
150 NEXT
160 GOTO 130
    
```

「`」は、キーボード左上にあります

- ① 「SAVE0」でプログラムをスロット 0 に保存
 - ② IchigoJam の電源を切って、電源ケーブル (micro USB) だけ残して他のケーブルを抜く
 - ③ IchigoJam の左側の黒いボタンを押しながら電源スイッチを入れる
- 以上の操作でプログラムが自動起動するので、IchigoJam を左右に振ってみましょう。

★30～110 行の 0,1 のパターンが、横にして見るとわかりますがハート形になっています。

(0=消灯、1=点灯) パターンを変えると、見える文字が変わります。

下の表で、0,1 のパターンを考えて、改造してみましょう。

行	30	40	50	60	70	80	90	100	110
データ									